

Setting up double integrals

Worksheet by ©Mike May, S. J, 2006.- maymk@slu.edu

The case of dx dy and dy dx

```
> restart;with(plots):
```

Overview

The first section of this worksheet provides a template for using Maple to set up and check double integrals symbolically.

The next section provides code for visualizing the region of integration. (Experience shows that students are more likely to have problems connecting a region of integration with limits of integration rather than with evaluating integrals.)

The third section deals with the problem of changing the order of integration.

```
>
```

Using Maple to check integration

This section provides a template for using Maple to check multiple integrals.

Much of this chapter deals with problems where you need to evaluate a multiple integral. It is worthwhile to note the Maple syntax used for multiple integrals, so that you can use Maple to check your work.

We do multiple integrals in Maple by nesting the integration commands.

It is useful to note that the command Int (upper case i) only sets up the integral. If you repeat the integral command with both versions of int you see that the integral is set up properly, as well as computing its value.

```
> Int(Int(x^2+y^2+x*y, x = 1..3), y=2..5);  
Int(int(x^2+y^2+x*y, x = 1..3), y=2..5);  
int(int(x^2+y^2+x*y, x = 1..3), y=2..5);
```

$$\int_2^5 \int_1^3 (x^2 + y^2 + xy) dx dy$$

$$\int_2^5 \left(\frac{26}{3} + 2y^2 + 4y \right) dy$$

146

(1.1)

Sometimes the integral is easier to evaluate in one order or the other.

```
> Int(Int(exp(x^2), x=y..1), y=0..1);  
Int(int(exp(x^2), x=y..1), y=0..1);  
int(int(exp(x^2), x=y..1), y=0..1);  
Int(Int(exp(x^2), y=0..x), x=0..1);  
Int(int(exp(x^2), y=0..x), x=0..1);  
int(int(exp(x^2), y=0..x), x=0..1);
```

$$\int_0^1 \int_y^1 e^{x^2} dx dy$$

$$\begin{aligned}
& \int_0^1 \left(\frac{1}{2} I \sqrt{\pi} \operatorname{erf}(Iy) - \frac{1}{2} I \operatorname{erf}(I) \sqrt{\pi} \right) dy \\
& \quad - \frac{1}{2} + \frac{1}{2} e \\
& \quad \int_0^1 \int_0^x e^{x^2} dy dx \\
& \quad \int_0^1 e^{x^2} x dx \\
& \quad - \frac{1}{2} + \frac{1}{2} e \tag{1.2}
\end{aligned}$$

We can also use the MultiInt command from the Student[MultivariateCalculus] package, specifying the output=steps option.

```

> with(Student[MultivariateCalculus]);
[ApproximateInt, ApproximateIntTutor, CenterOfMass, ChangeOfVariables,
CrossSection, CrossSectionTutor, DirectionalDerivative, DirectionalDerivativeTutor,
FunctionAverage, Gradient, GradientTutor, Jacobian, LagrangeMultipliers, MultiInt,
Revert, SecondDerivativeTest, SurfaceArea, TaylorApproximation,
TaylorApproximationTutor]

```

```

> MultiInt(exp(x^2), y=0..x, x=0..1, output=steps);

```

$$\begin{aligned}
& \int_0^1 \int_0^x e^{x^2} dy dx \\
& = \int_0^1 \left(e^{x^2} y \right) \Big|_{y=0}^{y=x} dx \\
& = \int_0^1 e^{x^2} x dx \\
& = \left(\frac{1}{2} e^{x^2} \right) \Big|_{x=0}^{x=1} \\
& \quad - \frac{1}{2} + \frac{1}{2} e \tag{1.4}
\end{aligned}$$

Use the template as needed.

```

>

```

▼ Visualizing the region of integration


The bigger problem in any section on multiple integration is not evaluating the integral. Rather it is to set the limits of integration correctly. This worksheet gives 2 blocks of code that can be used to see the region that goes with a set of limits of integration.

▼ Quick and dirty visualization - implicitplot

The fast and easy way to see a collection of curves in the x-y plane is to use the `implicitplot` command from the `plots` package. It lets you plot a series of curves without solving for x and y.


Consider integrating over the region bounded by the curves $y=x^2$ and $y=x$ between $x=0$ and $x=2$.

```
> implicitplot([y=x^2,y=x,x=0, x=2],x=-5..5,y=-5..5,axes=boxed);
```




We can make the output easier to understand if we add a list of colors to identify the graphs.

```
> implicitplot([y=x^2,y=x,x=0, x=2],x=-5..5,y=-5..5,axes=boxed,color=[red,blue,green,black],thickness=2);
```



From the graph, it is clear that we are interested in the region with x from 0 to 2 and y from 0 to 4.

```
> implicitplot([y=x^2,y=x,x=0, x=2],x=0..2,y=0..4,  
axes=boxed, color=[red,blue, green, black]);
```



Integrating with respect to y on the inside

The first block of code below considers the case when we integrate with respect to y on the inside. Thus x ranges from the left end of the region to the right end of the region, and for a particular x, the value of y ranges from a bottom curve to a top curve. The integral should end with $dydx$.

Initially we are only interested in setting up the integral.

```
> topcurve := x -> x^2;
   bottomcurve := x -> x;
   leftend := 0; rightend := 2;
   print("region for ", int(int(f(x,y),
     y=bottomcurve(x)..topcurve(x)), x=leftend..rightend));
      topcurve := x → x2
      bottomcurve := x → x
      leftend := 0
      rightend := 2
```


$$\text{"region for "}, \int_0^2 \int_x^{x^2} f(x, y) \, dy \, dx$$

(2.2.1)

It is important to note that the limits of integration work when we add in the "x=" and y=". Having the integrals correctly set up, we then check that the visualization works.

```
> topcurve := x -> x^2;
bottomcurve:= x -> x;
leftend := 0; rightend := 2;
print("region for ", int(int(f(x,y),
    y=bottomcurve(x)..topcurve(x)), x=leftend..rightend));
lines := {}:
for i from 0 to 10 do
    tempx := leftend + i/10*(rightend - leftend):
    lines := lines union
        {[[tempx, topcurve(tempx)], [tempx, bottomcurve
(tempx)]]};
od:
plots[display]
    ([plot({[x, topcurve(x), x=leftend..rightend],
        [x, bottomcurve(x), x=leftend..rightend]}),
    plot(lines, color=BLACK)]);
    topcurve := x -> x^2
    bottomcurve := x -> x
    leftend := 0
    rightend := 2
```

$$\text{"region for "}, \int_0^2 \int_x^{x^2} f(x, y) \, dy \, dx$$



Notice that the code does not check to see which curve is on top.

The graph has the lines that correspond to particular values of x . This should remind you of the volume of revolution problems where you cut an area into thin rectangles to justify using an integral.

Integrating with respect to x on the inside

The second block of code considers the case when we integrate with respect to x on the inside. Thus y ranges from the bottom end of the region to the top end of the region, and for a particular y , the value of x ranges from a left curve to a right curve. The integral should end with $dx dy$.

As above, we start by just trying to set up the integral, then looking at a longer block of code to see the integral and its visualization.

```
> leftcurve := y -> y^2 - 1;
   rightcurve:= y -> y + 1;
   bottomend := -1; topend := 2;
   print(`region for ` ,
         int(int(f(x,y),x=leftcurve(y)..rightcurve(y)), y=
         bottomend..topend));
           leftcurve:=y→y2 - 1
           rightcurve:=y→y + 1
```


$$\begin{aligned}
 & \text{bottomend} := -1 \\
 & \text{topend} := 2 \\
 & \text{region for } \int_{-1}^2 \int_{y^2-1}^{y+1} f(x,y) \, dx \, dy \qquad (2.3.1)
 \end{aligned}$$

Since the inside variable is x, the top inside limits only get to use the variable y.

```

> leftcurve := y -> y^2 - 1;
rightcurve := y -> y + 1;
bottomend := -1; topend := 2;
print(`region for ` ,
      int(int(f(x,y),x=leftcurve(y)..rightcurve(y)), y=
bottomend..topend));
lines := {}:
for i from 0 to 10 do
  tempy := bottomend + i/10*(topend - bottomend):
  lines := lines union
    {[[leftcurve(tempy), tempy], [rightcurve(tempy),
tempy]]};
od:
plots[display]([plot([leftcurve(y), y, y=bottomend..topend],
                    [rightcurve(y), y, y=bottomend..topend])
),
               plot(lines, color=BLACK)]);
leftcurve := y -> y^2 - 1
rightcurve := y -> y + 1
bottomend := -1
topend := 2
region for , \int_{-1}^2 \int_{y^2-1}^{y+1} f(x,y) \, dx \, dy

```



The lines in this case correspond to integrating for particular values of y ,

>

Exercises:

1) Define the limits of integration to define the double integral over the regions defined below. Modify the code above and check your work on 2 of those problems.

(a) A triangle with vertices at $(-1, 1)$, $(-1, -2)$, and $(3, -2)$.

>

(b) A triangle with vertices at $(0, 1)$, $(2, 1)$, and $(1, 3)$.

>

(c) A quadrilateral with vertices at $(1, 0)$, $(4, 1)$, $(4, 2)$, and $(1, 2)$.

>

2) For the integrals given below, sketch the region of integration and evaluate the integrals.

(a) $\int_1^3 \int_0^4 e^{x+y} dy dx;$

>

(b) $\int_0^2 \int_0^x e^{x^2} dy dx;$

$$(c) \int_{-2}^0 \int_{-\sqrt{9-x^2}}^0 2xy \, dy \, dx;$$

$$\text{int(int(2*x*y, y = -sqrt(9-x^2) .. 0), x = -2 .. 0);$$

14

(3.1)

Changing the order of integration

The book also has several problems where you are asked to switch the order of integration. In the exercise, this turns impossible integrals into simple ones. To switch from $dydx$ to $dx dy$, the boundary curves start in the form $y = f(x)$, and need to be converted to the form $x = g(y)$.

Consider, for example, the region with x between 0 and 4 bounded by $y=2x^2$ and $y = 4\sqrt{x}$. It is also the region with y between 0 and 8, bounded by the curves $x=\sqrt{y/2}$ and $x=(y/4)^2$.

We first set up the two double integrals.

```
> topcurve := x -> 4*sqrt(x);
bottomcurve := x -> x^2/2;
leftend := 0; rightend := 4;
print(`region for`, int(int(f(x,y),
y=bottomcurve(x)..topcurve(x)), x=leftend..rightend));
```

```
leftcurve := y -> sqrt(2*y);
rightcurve := y -> (y/4)^2;
bottomend := 0; topend := 8;
print(`region for`,
int(int(f(x,y), x=leftcurve(y)..rightcurve(y)), y=
bottomend..topend));
```

$$\text{topcurve} := x \rightarrow 4\sqrt{x}$$

$$\text{bottomcurve} := x \rightarrow \frac{1}{2}x^2$$

$$\text{leftend} := 0$$

$$\text{rightend} := 4$$

$$\text{region for, } \int_0^4 \int_{\frac{1}{2}x^2}^{4\sqrt{x}} f(x,y) \, dy \, dx$$

$$\text{leftcurve} := y \rightarrow \sqrt{2y}$$

$$\text{rightcurve} := y \rightarrow \frac{1}{16}y^2$$

$$\text{bottomend} := 0$$

$$\text{topend} := 8$$

$$\text{region for, } \int_0^8 \int_{\sqrt{2y}}^{\frac{1}{16}y^2} f(x,y) \, dx \, dy$$

(4.1)

Now we look at graphs with inner integration lines drawn in.

```

> topcurve := x -> 4*sqrt(x);
bottomcurve:= x -> x^2/2;
leftend := 0; rightend := 4;
print(`region for `, int(int(f(x,y),
    y=bottomcurve(x)..topcurve(x)), x=leftend..rightend));
lines := {};
for i from 0 to 10 do
    tempx := leftend + i/10*(rightend - leftend):
    lines := lines union
        {[[tempx,topcurve(tempx)], [tempx, bottomcurve(tempx)
]]};
od:
plots[display]
    ([plot({[x, topcurve(x),x=leftend..rightend],
        [x, bottomcurve(x),x=leftend..rightend]}),
    plot(lines, color=BLACK)];numlines := 10:

```

```

leftcurve := y -> sqrt(2*y);
rightcurve:= y -> (y/4)^2;
bottomend := 0; topend := 8;
print(`region for `,
    int(int(f(x,y),x=leftcurve(y)..rightcurve(y)), y=
bottomend..topend));
lines := {};
for i from 0 to 10 do
    tempy := bottomend + i/10*(topend - bottomend):
    lines := lines union
        {[[leftcurve(tempy), tempy], [rightcurve(tempy),
tempy]]};
od:
plots[display]([plot({[leftcurve(y),y,y=bottomend..topend],
    [rightcurve(y),y,y=bottomend..topend]}),
    plot(lines, color=BLACK)]);

```


$$topcurve := x \rightarrow 4\sqrt{x}$$

$$bottomcurve := x \rightarrow \frac{1}{2}x^2$$

$$leftend := 0$$

$$rightend := 4$$

$$region\ for\ ,\ \int_0^4 \int_{\frac{1}{2}x^2}^{4\sqrt{x}} f(x,y) dy dx$$




leftcurve := $y \rightarrow \sqrt{2y}$

rightcurve := $y \rightarrow \frac{1}{16}y^2$

bottomend := 0

topend := 8

region for , $\int_0^8 \int_{\sqrt{2}\sqrt{y}}^{\frac{1}{16}y^2} f(x,y) dx dy$



Notice that there is no easy formula for changing the limits on the region of integration. You need to look at the region, decide which curves are top and bottom, left and right, and solve for the appropriate variable.

$$\int_0^2 f(x) dx$$

Exercise:

3) For each of the integrals given below, reverse the order of integration and evaluate the integral. Use Maple to graph the region in both directions to check that you have the same region for both integrals.

(a) $\int_0^1 \int_y^1 e^{x^2} dx dy;$

[>

(b) $\int_0^3 \int_{y^2}^9 y \sin(x^6) dx dy;$

[>

(c) $\int_0^1 \int_{\sqrt{y}}^1 \sqrt{2+x^3} \, dx \, dy;$

[>
[>